



A situation with a constant rate of change can be modeled with a linear function: $f(x) = mx + b$, where b is the initial value and m is the rate of change. The graph of a linear function will be a straight line. In this module we will look at functions that are “almost” linear functions. These will be piece-wise linear functions; functions that are linear over an interval of the domain, but whose rate of change will vary from one interval of the domain to another interval on the domain. Graphically, our piecewise linear functions will look something like this:



We will focus on piecewise functions that are linear over each interval, but non-linear functions can also be created. Here is a graph of a piecewise function that is linear for values of x that are less than zero, but is quadratic for values of x that are greater than zero.



When would we use piecewise functions? Some situations have a rate of change that changes. Let's look at an example where there is a bulk discount. We need to print up a large number of flyers and go to the local copy shop. If they charge \$.05 per copy then a model for the amount charged for n copies would be $C(n) = .05n$. This is a linear function with a constant rate of change.

But most shops offer a discount on large orders; this shop charges \$.05 for orders up to 100, and then \$.03 for any copies after the first 100. One way to think about this would be:

If the number of copies ordered is ≤ 100	If the number of copies ordered is > 100 , the first 100 copies costs us \$5, then we are charged 0.03 for all copies beyond 100.
---	---

$$\text{Model } C(n) = 0.05n, \text{ for } n \leq 100$$


$$\text{Model } C(n) = 5 + 0.03(n - 100), \text{ for } n > 100$$

There is a way to combine these two models into one model

$$C(n) = \begin{cases} 0.05n, & \text{if } n \leq 100 \\ 5 + 0.03(n - 100), & \text{if } n > 100 \end{cases}$$

This called a piecewise (linear) function

Graphically, we have a function formed by pieces of the two formulas in our function. The domain of each piece is given by the “if” statement and that tells us when to use which part.




Note: Don't bother simplifying the expression $5 + 0.03(n - 100)$, Excel has no problem with it and it will be simpler to think of the problem by expressing the function in terms of: Amount Already Paid + New Rate *(The Number beyond the First Interval).

Let's talk about "if". "If" is an operation in logic, and in Excel. An "IF" statement might be along the lines of "If I do my homework then I will pass the class, otherwise I will fail."

The format has three parts: IF(something is true, then something happens, otherwise something else will happen)

A very simple piecewise function would be: $f(x) = \begin{cases} 0, & \text{if } x \leq 5 \\ 10, & \text{if } x > 5 \end{cases}$ For this function, if the input is equal or less than 5 the result will be 0, but if the input is greater than 5 the result will be 10.


To translate this function into a statement that Excel understands we will use an IF statement.



In general, with IF statements, it is easiest to work from left to right with the domain of the function, from smallest inputs to largest inputs. To use this function we have to first ask the question, is the input equal or less than 5? If the answer is yes, then our result is 0. If the answer is no, then the result is 10. Note, there is **no need** to ask a second question "is the input greater than 5?" If it is **not** equal or less than 5 we know it **must be** greater than 5.

Again, there are at least three parts to the IF statement in Excel. The first is called the logical test. This is where we check “is the input less than 5?” If it is true, then the second part is the value to assign. If it is false, then the third part is the value to assign.

So $f(x) = \begin{cases} 0, & \text{if } x \leq 5 \\ 10, & \text{if } x > 5 \end{cases}$ would be translated into Excel as $=IF(x \leq 5, 0, 10)$



	x	y
3	0	0
4	1	0
5	2	0
6	3	0
7	4	0
8	5	0
9	6	10
10	7	10
11	8	10
12	9	10
13	10	10

x	y
=0	=IF(B3<=5, 0, 10)
=B3+1	=IF(B4<=5, 0, 10)
=B4+1	=IF(B5<=5, 0, 10)
=B5+1	=IF(B6<=5, 0, 10)
=B6+1	=IF(B7<=5, 0, 10)
=B7+1	=IF(B8<=5, 0, 10)
=B8+1	=IF(B9<=5, 0, 10)
=B9+1	=IF(B10<=5, 0, 10)
=B10+1	=IF(B11<=5, 0, 10)
=B11+1	=IF(B12<=5, 0, 10)
=B12+1	=IF(B13<=5, 0, 10)

What about something more complicated?

$$f(x) = \begin{cases} 0, & \text{if } x \leq 5 \\ 10, & \text{if } 5 < x \leq 8 \\ 15, & \text{if } x > 8 \end{cases}$$

The basic format of an IF statement in Excel works very nicely with a two piece equation. If we have a three piece equation we will need to nest one IF statement within a second IF statement. Starts off the same:

$= IF(x \leq 5, y = 0, \text{then what?})$

If x is not ≤ 5 , then we know it is greater than 5. But what we don't know is whether x is less than or equal to 8, or whether x is greater than 8 so we need a second logical test and that is why we need a second IF statement:

$= IF(x \leq 5, y = 0, IF(x \leq 8, y = 10, \text{or else what?}))$
 $= IF(x \leq 5, y = 0, IF(x \leq 8, y = 10, 15))$

Note, why don't we have to use $IF(x \leq 5, y = 0, IF(5 < x \leq 8, y = 10, 15))$? We will

only reach the second IF statement if we fail the first test, so no need to check that $5 < x$. In fact, if you did use $IF(5 < x \leq 8, y = 10, 15)$ you would not get correct results as Excel does not understand compound inequalities.

Handy Tip: the number of “ifs” needed will be one less than the number of pieces in the piecewise function. Three pieces requires two ifs, a four piece function will require three ifs.

E	F	G	H	I
x	y			
0	0			
1	0			
2	0			
3	0			
4	0			
5	0			
6	10			
7	10			
8	10			
9	15			
10	15			

D	E	F	G
x		y	
=0		=IF(E3<=5, 0,IF(E3<=8,10,15))	
=E3+1		=IF(E4<=5, 0,IF(E4<=8,10,15))	
=E4+1		=IF(E5<=5, 0,IF(E5<=8,10,15))	
=E5+1		=IF(E6<=5, 0,IF(E6<=8,10,15))	
=E6+1		=IF(E7<=5, 0,IF(E7<=8,10,15))	
=E7+1		=IF(E8<=5, 0,IF(E8<=8,10,15))	
=E8+1		=IF(E9<=5, 0,IF(E9<=8,10,15))	
=E9+1		=IF(E10<=5, 0,IF(E10<=8,10,15))	
=E10+1		=IF(E11<=5, 0,IF(E11<=8,10,15))	
=E11+1		=IF(E12<=5, 0,IF(E12<=8,10,15))	
=E12+1		=IF(E13<=5, 0,IF(E13<=8,10,15))	

Now let's go back to the copy shop example, the function for the total cost is we order n copies of the flyer will be:

$$C(n) = \begin{cases} 0.05n, & \text{if } n \leq 100 \\ 5 + 0.03(n - 100), & \text{if } n > 100 \end{cases}$$

To start translating this function into an IF statement think of the function as:

IF $n \leq 100$, then use $c = .05n$, otherwise use $c = 5 + 0.03(n - 100)$

This becomes:

$$= IF(n \leq 100, .05n, 5 + 0.03(n - 100))$$

BUT, two more changes are needed. The variable n will need to be replaced with a cell reference to the input values, the number of copies ordered, AND the implied multiplication needs to become explicit:

$$= IF(B11 \leq 100, .05 * B11, 5 + 0.03 * (B11 - 100))$$

	A	B	C	D15	A	B	C	D	E	F
1	Example 2.4.1 A copy Shop charges \$0.05 per page up to 100 copies,									
2	Example 2.4.1 A copy Shop charges \$0.05 per page up to 100 copies, and then charges \$0.03 per page.									
3										
4	C(n) =	0.05n	if n<=100		C(n) =	0.05n	if n<=100			
5		5 + 0.03(n-100)	if n>100			5 + 0.03(n-100)	if n>100			
6										
7	Start	0			Start	95				
8	increment	10			increment	1				
9										
10		n, # of copies	total cost			n, # of copies	total cost			
11		=B7	=IF(B11<=100,0.05*B11,5+0.03*(B11-100))		0	95	\$ 4.75			
12		=B11+\$B\$8	=IF(B12<=100,0.05*B12,5+0.03*(B12-100))		1	96	\$ 4.80			
13		=B12+\$B\$8	=IF(B13<=100,0.05*B13,5+0.03*(B13-100))		2	97	\$ 4.85			
14		=B13+\$B\$8	=IF(B14<=100,0.05*B14,5+0.03*(B14-100))		3	98	\$ 4.90			
15		=B14+\$B\$8	=IF(B15<=100,0.05*B15,5+0.03*(B15-100))		4	99	\$ 4.95			
16		=B15+\$B\$8	=IF(B16<=100,0.05*B16,5+0.03*(B16-100))		5	100	\$ 5.00			
17		=B16+\$B\$8	=IF(B17<=100,0.05*B17,5+0.03*(B17-100))		6	101	\$ 5.03			
18		=B17+\$B\$8	=IF(B18<=100,0.05*B18,5+0.03*(B18-100))		7	102	\$ 5.06			
19					8	103	\$ 5.09			
20					9	104	\$ 5.12			
21					10	105	\$ 5.15			
22					11	106	\$ 5.18			
23					12	107	\$ 5.21			
24					13	108	\$ 5.24			

I adjusted my start and increment values to check that the cost per copy is increasing by 5¢ until 100 copies is reached, then it increases by 3¢

Taking our example one more step: we will get an additional discount if the order is more than 250 copies; those copies will only be \$0.02.

We can write the function out in words, then as a piecewise formula, then translate it into an If statement. (I know that sounds like extra work, but it will be helpful to take the time to do this. At least until you have more practice.)

If the number of copies is less than 100, we will be charged $0.05 \times$ number of copies. If not, but the number of copies is less than 250, then we will be charged 0.05×100 for the first copies, plus $0.03 \times$ the number of copies beyond the first 100, if more than 250 copies are ordered then we are charged 0.05×100 for the first 100, 0.03×150 for the next 150 copies plus $0.02 \times$ the number of copies beyond 250.

$$C(n) = \begin{cases} .05n, & \text{if } n \leq 100 \\ 5 + .03(n - 100), & \text{if } 100 < n \leq 250 \\ 5 + 4.5 + .02(n - 250), & \text{if } n > 250 \end{cases}$$

Note, take a moment and think about how many copies we will get for the 3¢ per copy price.

	A	B	C	D	E	F
1	Example 2.4.1 A copy Shop charges \$0.05 per page up to 100					
2	copies, and then charges \$0.03 per page up through 250 pages, and					
3	\$0.02 per page after that.					
4						
5	C(n) =	0.05n	if n<=100			
6		5 + 0.03(n-100)	if 100<n<=250			
7		9.50 + 0.02(n-250)	if n>250			
8						
9	Start	245				
10	increment	1				
11						
12		n, # of copies	total cost			
13		245	\$ 9.35			
14		246	\$ 9.38			
15		247	\$ 9.41			
16		248	\$ 9.44			
17		249	\$ 9.47			
18		250	\$ 9.50			
19		251	\$ 9.52			
20		252	\$ 9.54			
21		253	\$ 9.56			
22		254	\$ 9.58			

	A	B	C
4			
5	C(n) =	0.05n	if n<=100
6		5 + 0.03(n-100)	if 100<n<=250
7		9.50 + 0.02(n-250)	if n>250
8			
9	Start	0	
10	increment	10	
11			
12		n, # of copies	total cost
13		=B9	=IF(B13<=100,0.05*B13,IF(B13<=250,5+0.03*(B13-100),9.5+0.02*(B13-250)))
14		=B13+\$B\$10	=IF(B14<=100,0.05*B14,IF(B14<=250,5+0.03*(B14-100),9.5+0.02*(B14-250)))
15		=B14+\$B\$10	=IF(B15<=100,0.05*B15,IF(B15<=250,5+0.03*(B15-100),9.5+0.02*(B15-250)))
16		=B15+\$B\$10	=IF(B16<=100,0.05*B16,IF(B16<=250,5+0.03*(B16-100),9.5+0.02*(B16-250)))
17		=B16+\$B\$10	=IF(B17<=100,0.05*B17,IF(B17<=250,5+0.03*(B17-100),9.5+0.02*(B17-250)))
18		=B17+\$B\$10	=IF(B18<=100,0.05*B18,IF(B18<=250,5+0.03*(B18-100),9.5+0.02*(B18-250)))
19		=B18+\$B\$10	=IF(B19<=100,0.05*B19,IF(B19<=250,5+0.03*(B19-100),9.5+0.02*(B19-250)))
20		=B19+\$B\$10	=IF(B20<=100,0.05*B20,IF(B20<=250,5+0.03*(B20-100),9.5+0.02*(B20-250)))
21		=B20+\$B\$10	=IF(B21<=100,0.05*B21,IF(B21<=250,5+0.03*(B21-100),9.5+0.02*(B21-250)))
22		=B21+\$B\$10	=IF(B22<=100,0.05*B22,IF(B22<=250,5+0.03*(B22-100),9.5+0.02*(B22-250)))
23		=B22+\$B\$10	=IF(B23<=100,0.05*B23,IF(B23<=250,5+0.03*(B23-100),9.5+0.02*(B23-250)))

Once our function is written, we can generate a graph. If you see any odd jumps or jags in your graph you should go back and check your function again.

	A	B	C	D	E	F	G	H	I	J	K	L																																		
	C(n) =	$0.05n$ if $n \leq 100$ $5 + 0.03(n - 100)$ if $100 < n \leq 250$ $9.50 + 0.02(n - 250)$ if $n > 250$																																												
	Start	0																																												
	increment	10																																												
		<table border="1"> <thead> <tr> <th>n, # of</th> <th>total cost</th> </tr> </thead> <tbody> <tr><td>0</td><td>\$ -</td></tr> <tr><td>10</td><td>\$ 0.50</td></tr> <tr><td>20</td><td>\$ 1.00</td></tr> <tr><td>30</td><td>\$ 1.50</td></tr> <tr><td>40</td><td>\$ 2.00</td></tr> <tr><td>50</td><td>\$ 2.50</td></tr> <tr><td>60</td><td>\$ 3.00</td></tr> <tr><td>70</td><td>\$ 3.50</td></tr> <tr><td>80</td><td>\$ 4.00</td></tr> <tr><td>90</td><td>\$ 4.50</td></tr> <tr><td>100</td><td>\$ 5.00</td></tr> <tr><td>110</td><td>\$ 5.30</td></tr> <tr><td>120</td><td>\$ 5.60</td></tr> <tr><td>130</td><td>\$ 5.90</td></tr> <tr><td>140</td><td>\$ 6.20</td></tr> <tr><td>150</td><td>\$ 6.50</td></tr> <tr><td>160</td><td>\$ 6.80</td></tr> </tbody> </table>		n, # of	total cost	0	\$ -	10	\$ 0.50	20	\$ 1.00	30	\$ 1.50	40	\$ 2.00	50	\$ 2.50	60	\$ 3.00	70	\$ 3.50	80	\$ 4.00	90	\$ 4.50	100	\$ 5.00	110	\$ 5.30	120	\$ 5.60	130	\$ 5.90	140	\$ 6.20	150	\$ 6.50	160	\$ 6.80							
n, # of	total cost																																													
0	\$ -																																													
10	\$ 0.50																																													
20	\$ 1.00																																													
30	\$ 1.50																																													
40	\$ 2.00																																													
50	\$ 2.50																																													
60	\$ 3.00																																													
70	\$ 3.50																																													
80	\$ 4.00																																													
90	\$ 4.50																																													
100	\$ 5.00																																													
110	\$ 5.30																																													
120	\$ 5.60																																													
130	\$ 5.90																																													
140	\$ 6.20																																													
150	\$ 6.50																																													
160	\$ 6.80																																													

